Kétváltozós függvények

Karsai János
2009. március 11.

Feladat

Tekintsünk egy kétváltozós függvényt. Vizsgáljuk meg grafikusan, keressünk szélsőértékhelyeket, nyeregpontokat. Rajzoljuk fel a gradiens mezőt és a hozzá illeszkedő trajektóriákat.

Melléktermék

Stílusok, stylesheet - k használata; InputForm, StandardForm, TraditionalForm; grafikus utasítások opciói, programcsomagok betöltés, utólagos betöltéas hatásának kivédése

Definíció

$$f(x, y) := -2\sqrt{x^2 + y^2} + x^4 + y^2 - 0.4\,x.$$

```
f[x_, y_] := -2 Sqrt[x^2 + y^2] + x^4 + y^2 - 0.4 x
```

Ábrázolás

```
x1 = -1.2; x2 = 1.2;
y1 = -2; y2 = 2;
```

```
Plot3D[f[x, y], {x, x1, x2}, {y, y1, y2},
 AxesLabel → {x, y, z}]
```



```
ContourPlot[f[x, y], {x, x1, x2}, {y, y1, y2},
 AxesLabel → {x, y}, Contours → 50]
```



Szélsőértékek

Numerikus keresés

```
min1 = FindMinimum[f[x, y], {x, -0.5}, {y, -1}]
```

```
f[x, y] /. min1[[2]]
```

```
ICmin = {{-0.5, -1.}, {0.8, -0.7}, {-0.5, 1.}, {0.8, 0.7}};
```

```
Table[FindMinimum[f[x, y], {x, ICmin[[i, 1]]}, {y, ICmin[[i, 2]]}], {i, 1, Length[ICmin]}]
```

```
{{-0.99134, {x → -0.569593, y → -0.821927}}, {-1.5506, {x → 0.791425, y → -0.611266}},
 {-0.99134, {x → -0.569593, y → 0.821927}}, {-1.5506, {x → 0.791425, y → 0.611266}}}
```

```
ICmax = {{0.1, 0.1}};
```

```
Table[FindMaximum[f[x, y], {x, ICmax[[i, 1]]}, {y, ICmax[[i, 2]]}], {i, 1, Length[ICmax]}]
```

$\{\{-1.08987\times 10^{-8}, \{x \to 9.19333\times 10^{-10}, y \to -5.18462\times 10^{-9}\}\}\}$

Elméleti módszer

```
grad = D[f[x, y], {{x, y}}]
```

$$\left\{-0.4 + 4 x^3 - \frac{2 x}{\sqrt{x^2 + y^2}}, 2 y - \frac{2 y}{\sqrt{x^2 + y^2}}\right\}$$

```
plt0 = Show[ContourPlot[f[x, y], {x, x1, x2}, {y, y1, y2},
    AxesLabel → {x, y}, Contours → 50, ContourShading → False],
  ContourPlot[grad == {0, 0}, {x, x1, x2}, {y, y1, y2},
    AxesLabel → {x, y}]
 ]
```



Gradiens mező

```
Remove[VectorFieldPlot]
```

```
<< VectorFieldPlots`;
```

```
grad = D[f[x, y], {{x, y}}]
```

$$\left\{-0.4+4x^3-\frac{2x}{\sqrt{x^2+y^2}},\ 2y-\frac{2y}{\sqrt{x^2+y^2}}\right\}$$

```
pltgrad = VectorFieldPlot[-grad, {x, -1.1, 1.1},
  {y, -1.5, 1.5}, Axes → True, ScaleFactor → 0.4, PlotPoints → 20]
```




```
Show[plt0, pltgrad]
```



Gradiens mozgás pályája

```
grad = D[f[x, y], {{x, y}}]
```

$$\left\{-0.4+4x^3-\frac{2x}{\sqrt{x^2+y^2}},\ 2y-\frac{2y}{\sqrt{x^2+y^2}}\right\}$$

```
grad /. {x → x[t], y → y[t]}
```

$$\left\{-0.4+4x[t]^3-\frac{2x[t]}{\sqrt{x[t]^2+y[t]^2}},\ 2y[t]-\frac{2y[t]}{\sqrt{x[t]^2+y[t]^2}}\right\}$$

```
Thread[{x'[t], y'[t]} == (-grad /. {x → x[t], y → y[t]})]
```

$$\left\{x'[t]==0.4-4x[t]^3+\frac{2x[t]}{\sqrt{x[t]^2+y[t]^2}},\ y'[t]==-2y[t]+\frac{2y[t]}{\sqrt{x[t]^2+y[t]^2}}\right\}$$

```
gradeqn[x0_, y0_] = {x'[t] == (-grad /. {x → x[t], y → y[t]})[[1]],
   y'[t] == (-grad /. {x → x[t], y → y[t]})[[2]],
   x[0] == x0, y[0] == y0};
```

```
plttraj = ParametricPlot[
  Evaluate[{x[t], y[t]} /. NDSolve[gradeqn[-0.32, -2], {x[t], y[t]}, {t, 0, 10}]],
  {t, 0, 10}, PlotStyle → {Thickness[0.01], Red}]
```

```
Show[plttraj, plt0, pltgrad, PlotRange → All]
```



Ráadás : interaktív játék

```
Manipulate[
 Show[plt0, plt1, plt2,
  ParametricPlot[Evaluate[{x[t], y[t]} /. NDSolve[gradeqn[p0], {x[t], y[t]}, {t, 0, 10}]],
   {t, 0, 10}, PlotStyle → {Thickness[0.01], Red}]
 ], {p0, {-1.1, -1.5}, {1.1, 1.5}, Locator},
 Initialization :→ (Needs["VectorFieldPlots`"]; f[x_, y_] = -2 Sqrt[x^2 + y^2] + x^4 + y^2 - 0.4 x;
   x1 = -1.1; x2 = 1.1; y1 = -1.5; y2 = 1.5;
   grad = D[f[x, y], {{x, y}}];
   gradeqn[{u_, v_}] = {x'[t] == (-grad /. {x → x[t], y → y[t]})[[1]],
     y'[t] == (-grad /. {x → x[t], y → y[t]})[[2]],
     x[0] == u, y[0] == v};
   plt0 = ContourPlot[f[x, y], {x, x1, x2}, {y, y1, y2},
     AxesLabel → {x, y}, Contours → 50, ContourShading → False];
   plt1 = ContourPlot[grad == {0, 0}, {x, x1, x2}, {y, y1, y2},
     AxesLabel → {x, y}];
   plt2 = VectorFieldPlot[-grad, {x, x1, x2},
     {y, y1, y2}, Axes → True, ScaleFactor → 0.4, PlotPoints → 20]
  )]
```